



دانشگاه صنعتی خواجه نصیرالدین طوسی

دانشکده مهندسی برق

جزوه خودآموز HSPICE

استاد:

دکتر آسمانی

روند پیدایش و تکامل HSPICE :

Hspice که تکامل یافته مدل های Spice، Spice2G.6، Spice3، میباشد در سال ۱۹۸۱ برای اولین بار معرفی شد. در سال ۱۹۸۴ نسخه پیشرفته تر آن قادر به تحلیل مداری با ۵۰۰۰۰ گره بود. در سال ۱۹۸۷ بهینه سازی هایی برای افزایش سرعت ودقت شبیه سازی روی آن انجام گرفت ، به طوری که در سال ۱۹۸۸ سرعت آن به ۱۰ تا ۱۰۰ برابر افزایش یافت در حالیکه قادر بود مداری با ۱۰۰۰۰۰ ترانزیستور را شبیه سازی کند. هم اکنون می توان گفت HSPICE یکی از قدرتمندترین نرم افزارهای شبیه سازی مدار های الکترونیکی می باشد. و لذا می توان از این نرم افزار در دروس مربوط به مدار های مجتمع سود برد.

• **طریقه نصب HSPICE 2008 :**

میتوانید نرم افزار را از سایت دانشگاه کپی کنید:

U:\Engsoft\Hspice\hspice_VA_2008.03

۱- از پوشه ی (Hspice_VA_2008.03) فایل setup را اجرا می کنیم تا این که برنامه نصب شود.

۲- فایل Hspice.exe را از درون Crack به درون پوشه (C:\synopsys\Hspice_A-2008.03\Bin\) می ریزیم.

۳- فایل های License (hspice_2007-09.lic و license_hspice.dat) به درون پوشه (C:\Flexlm\) میریزیم.

۴- با انجام Right click روی my computer گزینه properties را انتخاب می کنیم تا با باز شدن صفحه system properties گزینه Advanced را انتخاب می کنیم و روی Click , Enviroment variables می کنیم.


۵- با باز شدن صفحه Enviroment variables در قسمت user variables , new را می زنیم.

۶- new user variable : در قسمت Variable name , LM_LICENSE_FILE را نوشته و در قسمت variable value C:\Flexlm\hspice_2007-09.lic را می نویسیم و Ok می کنیم.

۷- اکنون نرم افزار آماده کار می باشد:
Allprograms\Hspice_A-2008.03\Hspice-2008.03
را انتخاب می کنیم.

- شمای کلی نرم افزار:

با باز شدن صفحه نرم افزار icon های زیر قابل ملاحظه هستند:



Open : برای باز کردن فایل توصیف مدار با پسوند .sp.

Edit LL: برای مشاهده فایل متنی خروجی که در صفحه notepad باز می شود و برخی اطلاعات مربوط به نتایج فایل شبیه سازی شده ، پس از شبیه سازی در آن نوشته می شود.

Edit NL: با انتخاب این گزینه فایل توصیف مداری که در ابتدا از open باز کرده بودیم برای تغییر در دسترس قرار می گیرد.

Edit SL: یک سری اطلاعات از حجم فضای اشغال شده در طول شبیه سازی به ما می دهد.

Explore: پوشه مربوط به فایل شبیه سازی شده را باز می کند.

Exit: برای خروج از نرم افزار.

Simulate: دستور شبیه سازی فایل مدار

Avanwaves: نرم افزار جانبی برای مشاهده شکل موج های خروجی حاصل از simulation

Cscope, Waveview: برای نمایش شکل موج های خروجی که این نرم افزار های جانبی باید به صورت جدا نصب شوند.

• چگونه یک فایل **Netlist** بنویسیم:

برای نوشتن یک فایل قابل فهم برای Hspice ابتدا یک فایل متنی (Text) تولید می کنیم. و سپس پسوند آن را از TXT به SP تغییر می دهیم. داخل فایل را آن طور که در ادامه گفته می شود به توصیف مدار از طریق نام گره ها و عناصر بین آن گره ها ، و بیان تحلیل های مورد نظرمی پردازیم.

با گسترش و پیچیدگی روزافزون مدارهای الکترونیکی استفاده از کامپیوتر در تجزیه و تحلیل اینگونه مدارها از ضرورت برخوردار است. کامپیوتر کمک می‌کند تا بتوانیم بینش خود را در موارد زیر، درباره مدارهای الکترونیکی تعمیق دهیم:

- ۱- بررسی اثر تغییرات اجزاء مدار و محاسبه حساسیت رفتار مدار نسبت به آن (شامل مقاومت‌ها، سلفها، خازن‌ها، دیودها و ترانزیستورها و ...)
- ۲- در نظر گرفتن آثار غیر خطی المان‌ها در رفتار مدار
- ۳- بهینه کردن مدار با توجه به موارد فوق
- ۴- بررسی رفتار مدار از نظر نویز و اعوجاج‌ها و همچنین از نظر تجزیه و تحلیل در حوزه فرکانس و بدست آوردن طیف فرکانسی بر مبنای تجزیه و تحلیل فوریه بدون اینکه نیاز به استفاده از دستگاه‌های گران قیمتی مانند Noise Figure Meter و یا Spectrum Analyzer داشته باشیم.

برنامه HSpice

HSpice برنامه‌ای برای شبیه‌سازی مدارهای الکتریکی و الکترونیکی می‌باشد. این برنامه قابلیت تجزیه و تحلیل‌های زیر را در مدارهای الکتریکی و الکترونیکی دارد.

- ۱- تجزیه و تحلیل dc
- ۲- تجزیه و تحلیل گذرا
- ۳- تجزیه و تحلیل نویز
- ۴- تجزیه و تحلیل ac
- ۵- تجزیه و تحلیل فوریه

در برنامه HSpice مدارها می‌توانند شامل مقاومت، خازن، سلف (بدون تزویج متقابل و با تزویج متقابل) کلیدهای کنترل شده با جریان یا ولتاژ، منابع ولتاژ و جریان مستقل و وابسته و عناصر نیمه‌هادی مانند دیود، ترانزیستور، JFET، MESFET و MOSFET باشند. برنامه HSpice قادر است خطوط انتقال را هم بعنوان یک عنصر مداری بررسی نماید.

در استفاده از برنامه HSpice برای تجزیه و تحلیل مدارها چهار مرحله اساسی زیر را باید در نظر گرفت:

- ۱- تبدیل مدار به یک برنامه (file Netlist) قابل فهم برای HSpice
- ۲- تعیین نوع تجزیه و تحلیل مورد نیاز

۳- تعیین ورودی و خروجی های مورد نظر

۴- اجرا کردن برنامه و دیدن شکل موجهای نقاط مختلف مدار

تخصیص شماره یا اسم به گره

هر مدار شامل چند گره و چند شاخه و یک یا چند ورودی و یک یا چند خروجی خواهد بود. گرهها به دو صورت مشخص می شوند: با نام (حرف یا کلمه) یا با شماره (عدد). در نامگذاری گرهها بصورت غیر عددی، نام گرهها نباید بیشتر از ۱۶ کاراکتر باشد و حتماً با یکی از حروف الفبا آغاز گردد. در نامگذاری می توان از کاراکترهای زیر استفاده کرد.


! # \$ % * + - / < > [] _

گره های مدار در برنامه HSpice از شماره 0 تا $1E16-1$ شماره گذاری می شوند. گره 0 و GND و

GND! و همگی نمایانگر گره زمین هستند. ترتیب شماره گذاری بقیه گرهها مهم نیست. در

شماره گذاری گرهها می توان بجای اعداد از اسمهای مختلفی (جهت خواناتر شدن یا معنی دار شدن اسم گرهها) از جمله in1, in2, o, out3, insx, outsx و غیره استفاده کرد.

تمام گرهها باید از طریق یک مسیر dc قابل اتصال به زمین باشند. این مسیر می تواند از طریق مقامتها، منابع تغذیه، سلفها، دیودها و ترانزیستورها ایجاد شود. اگر در یک مدار گره ای وجود داشته باشد که هیچ مسیر dc به زمین نداشته باشد، می توان با اضافه نمودن یک یا چند مقاومت بزرگ به نقطه مناسب مدار آن مسیر را ایجاد کرد. بدیهی است که چون مقاومت اضافی بسیار بزرگ انتخاب می شود دخالتی در طرز کار مدار نخواهد داشت. برای مثال در شکل ۱، گره ۴ و یا ۳ هیچ مسیر dc به زمین ندارد. با اضافه نمودن یک مقاومت بزرگ موازی با خازن C2 این مسیر ایجاد می گردد.



شکل ۱

با استفاده از دستور GLOBAL می توان گره مورد نظر را در کل مدار مشترک کرد. اگر گرهی بصورت

GLOBAL تعریف شود آن گره در تمامی زیر برنامه ها و برنامه، یک گره در نظر گرفته می شود.

مثال:

در تعریف گره‌های داخل زیر مدار یا ماکرو به عنوان یک خروجی با نقطه متمایز می‌شوند. و به فرم کلی زیر بیان می‌شود.

نام گره. نام زیرمدار مثال: گره یک از زیر مدار X1 : X1.1

اگر در یک مدار بزرگ چند تا زیر مدار داشته باشیم و بخواهیم مقادیر ولتاژ یا جریان گره‌های داخل زیر مدار I را آدرس دهی کنیم باید به صورت زیر بنویسیم :

مثلا : ولتاژ گره ۱ از زیر مدار X1 در زیرمدار XAND V(XAND.X1.1)

تخصیص مقدار به المانها یا پارامترهای آنها

مقادیر اجزاء مدار در برنامه HSpice می‌توانند اعداد صحیح مثبت و منفی و یا اعشاری و یا نمایی باشند. برای مثال اعدادی مثل ۱۲ -۰/۰۱۲ -۴ E-۳/۱۴ قابل قبول‌اند. می‌توان از حروف لاتین بعنوان پسوند با اعداد برای بیان ضریب عدد استفاده کرد. جدول زیر حروف مربوطه و معادل آنها را نشان می‌دهد.

F	10^{-15}	femto	K		
P	10^{-12}	pico	MEG	10^6	Mega
N	10^{-9}	nano	G	10^9	Giga
U	10^{-6}	micro	T	10^{12}	tera
M	10^{-3}	mili			

تبصره: حرف m یا M برای mili است و برای 10^6 باید از MEG استفاده شود.

در HSpice حروف زیر برای نشان دادن واحدهای مشخص بکار می‌روند.

A: ampere	Hz: hertz	H: henry	DEG: degree
V: Volts	OHM: ohm	F: farad	

روش بیان المانها

در برنامه HSpice هر نوع عنصر یا المان مدار دارای یک اسم است که این اسم با یک حرف مشخص شروع می‌شود. جدول زیر اولین حرف در نامگذاری اجزاء مدار را برای المانهای مختلف مشخص می‌کند.

Symbols of Circuit Elements and Sources	
First Letter	Circuit Elements and Source
B	Buffer
C	Capacitor
D	Diode
E	Voltage Dependent Voltage source
F	Current- Dependent Current Source

G	Voltage Dependent Current Source
H	Current Dependent Voltage Source
I	Independent Current Source
J	JFET and MESFET Transistor
K	Mutual Inductor
L	Inductor
M	MOS Field Effect Transistor
Q	Bipolar Junction Transistor
R	Resistor
T, U, W	Transmission Line
V	Independent Voltage Source
X	Subcircuit call

اسم هر عنصر می تواند دارای ۱۶ کاراکتر مختلف باشد و از آن ۱۶ کاراکتر حرف اول حتماً باید مطابق جدول قبل انتخاب گردد بقیه کاراکترها می توانند شامل حرف و عدد باشند که بطور دلخواه انتخاب می گردند. در اسم گذاری تفاوتی بین حروف کوچک و بزرگ وجود ندارد.

فرمت نوشتن المانهای دو سر

بطور کلی فرمت نوشتن یک عنصر دو سر از یک مدار در برنامه HSpice به صورت شکل زیر است.

مقدار N+ N- نام المان

N+ و N- شماره دو گره ای هستند که المان بین آن دو گره قرار گرفته است و فرض بر اینست که

جریان از گره N+ به سمت گره N- جاری است.

بیان مدل برای المان

گاهی مقدار یک المان، تابع پارامترهای مختلفی است در این گونه موارد پس از تعیین محل المان در

مدار، اثر پارامترها را با یک "اسم مدل" و سپس نوع مدل را توسط دستور Model تعیین می کنند. در

این گونه موارد فرمت کلی بیان المان به شکل زیر است.

[مقدار] نام مدل N+ N- نام المان

برای بیان دقیق یک المان در HSpice از دستور MODEL استفاده می شود. می توان از "یک نام مدل" و

یک دستور MODEL برای چندین عناصر مشابه در مدار استفاده کرد. شکل کلی دستور MODEL به صورت

زیر است .

.MODEL MNAME TYPE P₁=X₁ P₂=X₂ ... P_n=X_n

MNAME اسمی انتخابی است که بعنوان "نام مدل" برای المان انتخاب می‌شود. TYPE مشخص کننده نوع المانی است که می‌خواهیم آنرا مدل کنیم. جدول زیر نحوه انتخاب TYPE را بیان می‌کند. P_1, P_2, \dots, P_n پارامترهای مربوط به المان و X_1, X_2, \dots, X_n به ترتیب مقادیر آن پارامترها می‌باشند.

Type name of elements	
Type name	Elements
AMP	operational amplifier model
C	capacitor model
CORE	magnetic core model
D	diode model
L	magnetic core mutual inductor model
NJF	n-channel JFET model
NMOS	n-channel MOSFET model
NPN	nnp BJT model
OPT	optimization model
PJF	p-channel JFET model
PLOT	plot model for the .GRAPH statement
PMOS	p-channel MOSFET model
PNP	pnnp BJT model
R	resistor model
U	lossy transmission line model (lumped)
W	lossy transmission line model
SP	S parameter

مثال :

D1 2 6 D914

.MODEL D914 D IS=1E-15 VJ=0.6 CJA=1.2E-13 CJP=1.3E-14
+ RS=1E+8 BV=70V

در اینجا دیود D1 بین گره‌های ۲ و ۶ قرار گرفته است و دارای "نام مدل" بنام D914 است. پس از اینکه برنامه HSpice این سطر را خواند دنبال سطری خواهد گشت که در آن مقادیر پارامترهای دیود مشخص شده باشد و آن سطر است که با دستور MODEL شروع می‌شود و در آن دستور دیود با نام مدل D914 تشریح شده است. حرف D برای تعیین type المان مدل شده (برای دیود) بکار رفته است.

منابع ولتاژ و جریان در HSpice

منابع ولتاژ و جریان در HSpice به دو دسته مستقل و وابسته تقسیم می‌شوند. منابع مستقل شامل منابع dc، منابع ac، منابع پالسی Pulse، منبع پاره‌خطی (PWL) Piece Wise Linear، منبع FM (مدوله شده با یک tone)، منبع AM، منبع چند جمله‌ای و منبع نمائی و منبع Mixed می‌باشند.

منابع مستقل

منبع ولتاژ یا جریان dc

منبع ولتاژ و جریان DC در HSpice به ترتیب با صورت کلی زیر بیان می‌شوند.

مقدار V_x N+ N- Dc یا مقدار V_x N+ N- Dc

مقدار I_x N+ N- Dc یا مقدار I_x N+ N- Dc

N+ و N- به ترتیب سر مثبت و منفی منبع می‌باشند. نمایش منابع جریان نیز مانند منابع ولتاژ می‌باشد با این تفاوت که واحدها برحسب آمپر و اسم منابع با I شروع می‌شود.

مثال برای منبع ولتاژ: $VCC\ 15\ 0\ DC\ 10V$ یعنی منبع ولتاژ VCC بین گره ۱۵ و زمین قرار گرفته و مقدار آن 10V است. کلمه DC قابل حذف است ضمناً می‌توان بصورت $DC=10$ نیز نوشت.

مثال برای منبع جریان: $Ims\ 15\ 0\ DC\ 10mA$ یعنی منبع جریان Ims بین گره ۱۵ و زمین قرار گرفته و مقدار آن 10mA است.

منبع ولتاژ یا جریان AC

این نوع منابع در تجزیه و تحلیل AC و نویز بکار می‌روند منبع ولتاژ و جریان AC در HSpice به ترتیب با صورت کلی زیر بیان می‌شود.

فاز مقدار V_x N+ N- AC

فاز مقدار I_x N+ N- ac

مثلاً برای بیان $V_x = A \sin(\omega t + \phi)$ در بین گره‌های ۱ و زمین بصورت زیر می‌نویسیم:

$V_x\ 1\ 0\ ac = A, \phi$ or $V_x\ 1\ 0\ ac\ A\ \phi$

منبع ولتاژ یا جریان AC با آفست DC

اگر منبعی دارای مقدار AC و DC داشته باشد به صورت زیر بیان می‌شود.

فاز مقدار V_x N+ N- DC مقدار AC

فاز مقدار I_x N+ N- DC مقدار AC

مثلا برای بیان $Vx = A + B \sin(\omega t + q)$ در بین گرههای ۱ و زمین بصورت زیر می نویسیم :

$Vx \ 1 \ 0 \ DC = A \ ac = B, q \ or \ Vx \ 1 \ 0 \ DC \ A \ ac \ B \ q$

منبع ولتاژ یا جریان سینوسی

این نوع منابع برای تجزیه و تحلیل گذرای (transient) مدارها استفاده می شوند. و به ترتیب دارای شکل کلی زیر می باشند.

$Vx \ N+ \ N- \ SIN \ V_0 \ V_A \ FREQ \ TD \ ALPHA \ TETA$

$Ix \ N+ \ N- \ SIN \ I_0 \ I_A \ FREQ \ TD \ ALPHA \ TETA$

N+ و N- شماره دو گره ای هستند که منبع Vx یا Ix بین آن دو گره قرار گرفته است. عبارتهای بالا به

ترتیب از روی روابط زیر نوشته شده اند.

$$V(t) = V_0 + V_A e^{-a(t-td)} \sin[2\pi f(t-td) + q]$$

$$I(t) = I_0 + I_A e^{-a(t-td)} \sin[2\pi f(t-td) + q]$$

که در معادله ولتاژ V_0 سطح ولتاژ DC منبع سینوسی، V_A دامنه منبع سینوسی، $f = FREQ$ فرکانس، $TD = td$ تأخیر زمانی، $\alpha = ALPHA$ ضریب تضعیف دامنه و $TETA = \theta$ فاز آن می باشند.
مثال:

$Vin \ 1 \ 0 \ Sin \ 0 \ 1v \ 1kHz$

که معرف یک منبع ولتاژ سینوسی بین گره ۱ و زمین با دامنه ۱ ولت و فرکانس ۱ کیلو هرتز (kHz) می باشد.

تبصره: اگر مقادیر TD و ALPHA و TETA در رابطه تعیین نگردند، Default آنها صفر خواهد بود.

مثال دیگر: فرض کنید بخواهیم منبع $V(t) = 1 + 10 \sin(2\pi \times 10^4 + 30^\circ)$ (بین گره ۳ و ۴) را در برنامه

بنویسیم این منبع بصورت زیر نوشته می شود.

$Vin \ 3 \ 4 \ Sin \ 1V \ 10V \ 1e4Hz \ 0 \ 0 \ 30$

و یا اگر بخواهیم منبع $V(t) = 2 - 3e^{-10(t-20 \times 10^{-6})} \sin(2\pi \times 10^4(t-20 \times 10^{-6}))$ (بین گره ۱۰ و

زمین) را بنویسیم این منبع بصورت زیر نوشته می شود.

$Vin \ 10 \ 0 \ Sin \ 2 \ 3 \ 10kHz \ 20US \ 10 \ 180$

منبع ولتاژ یا جریان پالسی

این نوع منابع برای تجزیه و تحلیل گذرای (transient) مدارها استفاده می شوند. و به ترتیب دارای شکل کلی زیر می باشند .

$Vx \ N+ \ N- \ Pulse \ V_1 \ V_2 \ TD \ TR \ TF \ PW \ PER$

$Ix \ N+ \ N- \ Pulse \ I_1 \ I_2 \ TD \ TR \ TF \ PW \ PER$

مثال

Vin 1 0 Pulse 0 5V 5uS 5uS 10uS 500uS 1ms

یعنی منبع ولتاژ پالسی بین گره‌های ۱ و صفر که دارای ولتاژ اولیه صفر بوده، پس از تأخیر $T_D = 5\mu\text{Sec}$ به

۵ ولت می‌رود و به اندازه $T_W = 500\mu\text{Sec}$ در ۵ ولت باقی می‌ماند. دارای $\text{Rise time} = 5\mu\text{sec}$ ، Fall time

$= 10\mu\text{s}$ و پریود 1mSec است .

منبع ولتاژ یا جریان پاره‌خطی (Piece Wise Linear)

این منابع بصورت شکل کلی زیر می‌باشند:

Vx N+ N- PWL T₁ V₁ T₂ V₂ T₃ V₃ ... T_n V_n R T_d

Ix N+ N- PWL T₁ I₁ T₂ I₂ T₃ I₃ ... T_n I_n R T_d


منبع ولتاژ در لحظه T_i دارای ولتاژ V_i می‌باشد و در بین دو زمان متوالی مقدار آن از خط اتصال بین دو

نقطه (T_i, V_i) و (T_{i+1}, V_{i+1}) تبعیت می‌کند . R برای نشان دادن تعداد دفعات تکرار موج بین زمانهای T_d تا

T_n (برای بعد از زمان T_n) می‌باشد نکته اینکه T_d حداقل باید از T_n کوچکتر باشد. برای مثال منبع شکل ۲

را می‌توان به صورت زیر تعریف نمود :

Vin 1 0 PWL 0 0 8us 24v 16us 32v 28us 32v 34us 12v 44us 0v



شکل ۲

منبع ولتاژ یا جریان نمایی


این نوع منابع به ترتیب دارای شکل کلی زیر می‌باشند.

Vx N+ N- EXP V₁ V₂ TD1 RTC TD2 FTC

Ix N+ N- EXP I₁ I₂ TD1 RTC TD2 FTC

که در روابط فوق TD1 زمان تاخیر اولیه، RTC = Rise time constant، TD2 زمان تاخیر، FTC = Fall time constant، V1 ولتاژ شروع و V2 ولتاژ انتهائی می باشند.

مثال شکل ۳: VSS N+ N- EXP 0 10V 5us 20us 100us 30us



شکل ۳

منبع ولتاژ یا جریان با مدولاسیون FM

این منبع از روی رابطه زیر تعریف می گردد و بیانگر یک موج سینوسی مدوله شده با یک فرکانس می باشد.

$$V(t) = V_0 + V_A \sin[2\pi f_c t + M \sin 2\pi f_s t]$$

شکل کلی این نوع منابع به صورت زیر است.

Vx N+ N- SFFM V0 VA FC Mdi Fs

Mdi همان M یا ضریب مدولاسیون است.

مثال :

Vsx 2 0 SFFM 0 1V 30MEG Hz 15 5kHz

ضریب مدولاسیون ۱۵ درصد است. منبع جریان با مدولاسیون FM نیز بصورت فوق تعریف می شود با این تفاوت که در منبع جریان بجای V از I استفاده می شود.

ISX 2 0 SFFM 0 1mA 30MEG Hz 15 5kHz

منبع ولتاژ یا جریان با مدولاسیون AM

این منبع بر مبنای رابطه زیر که بیانگر یک موج سینوسی است که با یک موج تک فرکانس به طور فاز مدوله شده است تعریف می گردد.

$$V(t) = S_a \{O_C + \sin[2\pi f_m(t - T_d)] \sin 2\pi f_c(t - T_d)\}$$

شکل کلی این نوع منابع به صورت زیر است.

Vx N+ N- AM Sa Oc Fm Fc Td

مثال :

VSX 2 0 AM 10 1V 5kHz 30MEG Hz 1m

برای منبع جریان نیز بصورت فوق تعریف می شود با این تفاوت که در منبع جریان بجای V از I استفاده می شود.

ISX 2 0 AM 10 1ma 5kHz 30MEG Hz 1m

منبع ولتاژ چند جمله ای (Poly)

منبع ولتاژی است که به بصورت یک تابع چند جمله ای از ولتاژ یک یا چند المان بوده و دارای سه

حالت زیر می باشد.

Poly(1) One-dimensional equation

Poly(2) Two-dimensional equation

Poly(3) Three-dimensional equation

شکل کلی این منابع بصورت زیر می باشد .

$$Y = P_0 + P_1 A + P_2 A^2 + \dots + P_n A^n$$

فرم نوشتن رابطه فوق در برنامه HSpice به صورت زیر است.

Vx N+ N- Poly NC+ NC- P0 P1 P2 ... Pn

که در این رابطه NC+ و NC- گره هایی هستند که منبع ولتاژ A (و یا شاخه حامل جریان) بین آن دو گره قرار گرفته است. مثلاً:

Vx N+ N- Poly(1) 10 0 0 2

که معرف $Y=2V(10)$ است. شکل های پیچیده تری هم برای اینگونه منابع قابل تعریف است.

مثال Poly(2)

VSX 1 2 Poly(2) 2 3 1 2 1

که معرف $Y = 1 + 2[V(2,3)] + [V(2,3)]^2$ است.

منبع ولتاژ یا جریان Mixed

این نوع منابع برای بیش از یک نوع تجزیه و تحلیل (مثلاً برای تجزیه و تحلیل AC، نویز، تجزیه و تحلیل گذرا) بکار می روند و به ترتیب دارای شکل کلی زیر می باشند.

Vin N+ N- Vdc AC Vac Sin Vo A f.

Iin N+ N- Idc AC Iac Sin Io A f.

در رابطه بالا f فرکانس، A دامنه و V_o مقدار آفست موج سینوسی (برای تجزیه و تحلیل گذرا)، V_{ac} دامنه و V_{dc} مدار DC موج AC (برای تجزیه و تحلیل AC و نویز) می‌باشند.
مثال

VIN 13 2 0.5 AC 1 SIN 0 1 1MEG

منابع وابسته

چهار نوع منبع وابسته در HSpice تعریف می‌گردد که به قرار زیرند.

- | | |
|-------------------------------|----------------|
| 1- منبع ولتاژ وابسته به ولتاژ | < نام منبع > E |
| 2- منبع ولتاژ وابسته به جریان | < نام منبع > H |
| 3- منبع جریان وابسته به ولتاژ | < نام منبع > G |
| 4- منبع جریان وابسته به جریان | < نام منبع > F |

منبع ولتاژ وابسته به ولتاژ

این نوع منابع برای شبیه سازی Op-Amp ایده‌ال، ترانس ایده‌ال، گیت‌های منطقی، المان تاخیر، اسیلاتور کنترل شونده با ولتاژ، منبع ولتاژ پاره خطی، چند جمله‌ای و منبع ولتاژ در سطح رفتاری بکار می‌رود. دو فرم از این منابع بصورت زیر می‌باشند.

E(name) N+ N- NC+ NC- <<MAX>=Val> <<MIN=Val> gain
E(name) N+ N- TRANSFORMER NC+ NC- k

$N+$ و $N-$ دو سر منبع ولتاژ وابسته و $NC+$ و $NC-$ سرهای ولتاژ کنترل کننده می‌باشند.

مثال :

Eop_amp 2 3 14 1 MAX=+5 MIN=-5 200
Etrans out 0 TRANSFORMER in 0 10

مثالهای قبلی به ترتیب برای Op-Amp ایده‌ال و TRANSFORMER ایده‌ال با تابع $V(out) = V(in)/10$ استفاده می‌شوند.

منبع ولتاژ وابسته به جریان

این نوع منابع برای شبیه سازی گیت‌های منطقی، المان تاخیر، منبع ولتاژ پاره خطی چند جمله‌ای و منبع ولتاژ در سطح رفتاری بکار می‌رود.

H(name) N+ N- VCUR <<MAX>=Val> <<MIN=Val> gain

N+ و N- سرهای مثبت و منفی منبع ولتاژ وابسته و VCUR منبع ولتاژی است که جریان آن (I)، اندازه ولتاژ منبع ولتاژ را کنترل می کند.
مثال:

H1 20 10 VCUR MAX=+10 MIN=-10 1000

منبع جریان وابسته به ولتاژ

این نوع منابع برای شبیه سازی سوئیچها، خازن متغیر با ولتاژ، منبع جریان پاره خطی چند جمله ای و منبع جریان در سطح رفتاری بکار می رود. فرم عمومی این منبع بصورت زیر است .

G (name) N+ N- NC+ NC- <<MAX>=Val> <<MIN=Val> gain

N+ و N- سرهای مثبت و منفی منبع جریان است و NC+ و NC- سرهای ولتاژ کنترل کننده می باشند .
مثال:

GA 12 10 3 4 500

منبع جریان وابسته به جریان

فرم متعارف این منابع بصورت زیر می باشد.

F(name) N+ N- VN <<MAX>=Val> <<MIN=Val> gain

در رابطه فوق N+ و N- سرهای مثبت و منفی منبع جریان است و VN منبع ولتاژ مفروضی است که جریان آن، اندازه منبع جریان خروجی را کنترل می کند.
مثال:

F1 10 5 Vin 10
F2 13 5 VSENS MAX=+3 MIN=-3 5

نمایش کامل عناصر غیرفعال

مقاومت

مقاومتها دارای دو شکل کلی زیر می باشند:

R(name) N+ N- < مقدار > (الف) :

R(name) N+ N- Model Name (RT0) مقدار نامی (ب) :

R3 2 3 10k یا R1 1 2 100hm مثال برای حالت الف :

R3 3 4 MSRES 10k مثال برای حالت ب :

.Model MSRES R R=1 TC1=0.02 TC2=0.002

پارامترهای مربوط به یک مقاومت بر اساس رابطه زیر تعریف می‌شوند.

$$R = R_{T0} \times (R \text{ ضریب}) \times [1 + TC1(T - T_0) + TC2(T - T_0)^2 + \dots]$$

R_{T0} مقدار مقاومت در درجه حرارت اتاق (27°C) است. پارامتر ضریب را در دستور Model با R نشان می‌دهند و معمولاً برابر با ۱ است. TC_1 و TC_2 و ... ضرایب حرارتی مقاومت (مرتبه اول و مرتبه دوم و ...) می‌باشند. جدول زیر پارامترهای مدل را برای یک مقاومت نشان می‌دهد و اگر در دستور Model هر کدام از پارامترها تعیین نشوند برنامه بطور اتوماتیک مقدار default آنها را در نظر می‌گیرد.

Model Parameters for Resistors			
Name	Meaning	Unit	Default
R	Resistance multiplier	-	1
TC1	First-order temperature coefficient	$1/C^\circ$	0
TC2	Second-order temperature coefficient	$(C^\circ)^{-2}$	0

خازن

شکل کلی خازنها به یکی از صورت‌های زیر است.

C (name) N+ N- مقدار

C (name) N+ N- Model Name مقدار IC= V_0

مثال برای حالت اول

CBY 3 4 10.7uF

مثال برای حالت دوم:

CBY 4 0 MSCAP 10UF IC=5V

.MODEL MSCAP C C=1 VC1=0.01 VC2=0.002 TC1=0.01

در مثال اخیر، $IC=5V$ شرایط اولیه خازن را بیان می‌کند (Initial Condition). پارامترهای مدل برای

خازنها بر اساس رابطه زیر بیان می‌شوند.

$$C = (\text{مقدار خازن}) \times (C \text{ ضریب}) \times [1 + VC_1 \times V_0 + VC_2 \times V_0^2] [1 + TC_1(T - T_0) + TC_2(T - T_0)^2]$$

که در آن VC_1 و VC_2 ضرایب ولتاژی و TC_1 و TC_2 ضرایب حرارتی خازن می‌باشند و V_0 پتانسیل

اولیه خازن است که با $IC=V_0$ تعریف و تعیین می‌شود. جدول بعد پارامترهای مدل را برای یک خازن نشان

می‌دهد. اگر در دستور Model هر کدام از پارامترها تعیین نشوند برنامه بطور اتوماتیک مقدار default آنها را در نظر می‌گیرد.

Model Parameters for Capacitors			
Name	Meaning	Unit	Default
C	Capacitance multiplier	–	1
VC1	First-order Voltage coefficient	Volts ⁻¹	0
VC2	Second-order Voltage coefficient	Volts ⁻²	0
TC1	First-order temperature coefficient	1/C ^o	0
TC2	Second-order temperature coefficient	(C ^o) ⁻²	0

سلف

شکل ساده بیان سلفها بصورت زیر می‌باشد:

L (name) N+ N- مقدار

L (name) N+ N- Model Name مقدار IC=I₀

I₀ بیانگر جریان اولیه سلف می‌باشد. مثال :

L1 2 3 SELF 0.1UH IC=-20mA

.Model SELF L L=1 IL1=0.1 IL2=0.002 TC1=0.02 TC2=0.001

پارامترهای مدل برای سلفها بر اساس رابطه زیر بیان می‌شوند.

$$L = (\text{مقدار سلف}) \times (\text{ضریب L}) \times [1 + IC_1 \times I_0 + IC_2 \times I_0^2] [1 + TC_1(T - T_0) + TC_2(T - T_0)^2]$$

جدول زیر پارامترهای مدل را برای سلفها نشان می‌دهد و اگر در دستور Model هر کدام از پارامترها تعیین نشوند برنامه بطور اتوماتیک مقدار default آنها را در نظر می‌گیرد.

Model Parameters for Inductors			
Name	Meaning	Unit	Default
L	Inductance multiplier	–	1
IC1	First-order Current coefficient	Amper ⁻¹	0
IC2	Second-order Current coefficient	Amper ⁻²	0
TC1	First-order temperature coefficient	1/C ^o	0
TC2	Second-order temperature coefficient	(C ^o) ⁻²	0


سلفهای کوپلاژدار

شکل کلی این نوع سلفها به صورت زیر می باشد.

ضریب کوپلاژ < نام سلف دوم > L < نام سلف اول > L < نام > K

مثال :


LA	3	4	10UH
LB	2	1	40UH
K1	LA	LB	0.9



شکل ۴

اگر جای نقطه یکی از سلفها مثلاً LB عوض شود و به صورت شکل ۵ درآید می توانیم بنویسیم.


LA	3	4	10UH
LB	1	2	40UH
K1	LA	LB	0.9



شکل ۵

یک ترانس با سر وسط و با ضریب کوپلاژ ۰/۹۹۹ (شکل ۶) به صورت زیر مدل سازی می شود.

L1	1	2	0.8mH
L2	3	4	0.5mH
L3	5	6	0.5mH
K12	L1	L2	0.999
K13	L1	L3	0.999
K23	L2	L3	0.999



شکل ۶

در صورت یکسان بودن ضریب کوپلاژ، می توان سه عبارت اخیر را بطور خلاصه در یک دستور نوشت:

KAL L1 L2 L3 0.999

معرفی و مدل سازی و کاربرد عناصر نیمه هادی در برنامه HSpice

بطور کلی عناصر نیمه هادی را به شکل های کلی زیر در برنامه HSpice می نویسند.

D(name) N+ N- Model name <AREA> دیود -۱

N+ گره آند ، N- گره کاتد و AREA سطح دیود می باشند .

مثال برای دیود :

```
D1 7 8 DIN4001
.Model DIN4001 D LEVEL=1 XP=0.0 EG=1.1 XOI=0.0 XOM=0.0
+XM=0.0 WP=0.0 WM=0.0 LP=0.0 LM=0.0 AF=1.0 JSW=0.0
+PB=0.65 PHP=0.8 M=0.2994 FC=0.95 FCS=0.4 MJSW=0.5 TT=2.446e-9
+BV=4.65 RS=19 IS=1.485e-11 CJO=1.09e-9 CJP=0.0 PJ=0.0 N=1.615
+IK=0.0 IKR=1.100e-2 IBV=2.00e-2
```

Q(name) NC NB NE Model <AREA> BJT ترانزیستور -۲

Q(name) NC NB NE NS Model name <AREA>

Model name NC, NB, NE و NS به ترتیب شماره گره های کلکتور، بیس، امیتر و زیرینا می باشند.

برای این نوع ترانزیستورها NPN یا PNP می باشد.

مثال :

```
Q(name) 1 2 3 T2n2222a
.Model T2n2222a NPN LEVEL=1 ISS=0. XTF=1 NS=1.00000
+CJS=0 VJS=0.5 PTF=0 MJS=0 EG=1.10000 AF=1 ITF=0.5
+VTF=1 BR=40 IS=1.6339e-14 VAF=103.40529 VAR=17.77498
+IKF=1 NE=1.31919 IKR=1 ISC=3.6856e-13 NC=1.10024 IRB=4.3646e-05
+NF=1.00531 NR=1.00688 RBM=1.0000e-02 RB=71.82988 RC=0.42753
+RE=3.0503e-03 MJE=0.32339 MJC=0.34700 VJE=0.67373 VJC=0.47372
+TF=9.693e-10 TR=380.00e-9 CJE=2.6734e-11 CJC=1.4040e-11 FC=0.950
+XCJC=0.94518
```

۳- ترانزیستور JFET و MESFET

J(name) ND NG NS Model name

ND, NG, NS به ترتیب شماره گره های درین، گیت و سورس می باشند. Model name برای این

نوع ترانزیستورها NFET یا PFET می باشد.

۴- ترانزیستور MOSFET

M(name) ND NG NS NB Model name L W AD PD AS PS

ND, NG, NS, NB, AD, PD, AS و PS به ترتیب شماره گره‌های درین، گیت، سورس، بدنه (Substrate)، مساحت درین، محیط درین، مساحت سورس و محیط سورس می‌باشند. Model name برای این نوع ترانزیستورها NMOS یا PMOS می‌باشد.

مثال :

M1 3 1 2 3 MOSN L=2u W=6u AD=36p PD=24u AS=36p PS=24u

اسم مدل MOSN می‌باشد که باید با دستور Model. تعریف شود.

بقیه پارامترهای المانهای مدل‌دار از روی مدل تعیین شده خوانده می‌شود. در هر مدلی هر پارامتری دارای معنی خاص می‌باشد ولی در اکثر مدلها این پارامترها یکسان و با یک اسم بیان شده‌اند. جهت تعیین پارامترهای تمام المانهای فوق از دستور MODEL استفاده می‌شود.

انواع متغیرها به عنوان خروجی

در هر مدار و ولتاژها و یا جریانهای مختلفی را می‌توانیم به عنوان متغیرهای خروجی تعریف کنیم.

شکل کلی تعریف متغیرهای خروجی به همراه مثالی در زیر ارائه شده است :

ولتاژ یک گره نسبت به زمین: (شماره گره) V (مثال : V(7))

اختلاف ولتاژ دو گره نسبت به هم : V(N₁,N₂) (مثال : V(3,4) یا V(3 4))

اختلاف پتانسیل دو سر یک المان دو سر : (نام المان دو سر) V (مثال : V(D1) یا V(R1))

پتانسیل یک سر از یک عنصر سه سر نسبت به زمین : (اسم المان) نام ترمینال عنصر سه سر V

مثال : پتانسیل درین ترانزیستور M₃ نسبت به زمین VD(M3)

اختلاف پتانسیل دو سر از یک عنصر سه سر : (اسم المان) نام دو ترمینال از عنصر سه سر V

مثال : VCE در ترانزیستور Q₄ VCE(Q4)

مشخص نمودن جریانها نیز مثل ولتاژها است. جریان جاری در یک المان در یک شاخه : (نام شاخه) I

مثال : I(VX) یا I(R₁) یا I(D₁)

جریان ورودی به ترمینال x از یک عنصر سه سر: (نام) IX (مثال : IB(Q2) یا IC(Q3))

که اولی جریان ورودی به کلکتور ترانزیستور Q₃ است و دومی جریان ورودی به بیس ترانزیستور Q₂ است.

در تجزیه و تحلیل AC با اضافه نمودن پسوند مناسب به نام ولتاژها و جریانها می‌توان اندازه، فاز،

بخش حقیقی و یا بخش موهومی یک متغیر را به دست آورد.

مثال :

VM(5)	اندازه ولتاژ گره ۵
VM(4,2)	اندازه ولتاژ بین گره‌های ۴ و ۲
VP(4,2)	اندازه فاز بین گره‌های ۴ و ۲
VDB(6)	اندازه ولتاژ گره ۶ برحسب dB
VCM(Q2)	اندازه ولتاژ کلکتور ترانزیستور Q ₂
VDSP(M6)	اندازه فاز VDS در ترانزیستور M6
VR(2,3)	بخش حقیقی اختلاف پتانسیل بین گره‌های ۲ و ۳
VI(2,3)	بخش موهومی اختلاف پتانسیل بین گره‌های ۲ و ۳

برای جریان نیز می‌توان تعاریف فوق را داشت.

IM(D1)	اندازه دامنه جریان در دیود D1
IP(D1)	اندازه فاز جریان در دیود D1
IR(VIN)	بخش حقیقی جریان منبع ولتاژ VIN

اگر در یک مدار بزرگ چند تا زیرمدار داشته باشیم و بخواهیم مقادیر ولتاژ یا جریان گره‌های داخل زیرمدار I را آدرس دهی کنیم باید به صورت زیر عمل کنیم :

ولتاژ گره ۱ از زیر مدار X1 از زیرمدار XAND (X1.1) V(XAND.X1.1)

انواع تجزیه و تحلیل مدار در برنامه HSpice

پنج نوع تجزیه و تحلیل مدار در برنامه HSpice قابل اجراست:

۱- تجزیه و تحلیل dc

۲- تجزیه و تحلیل ac

۳- تجزیه و تحلیل گذرا

۴- تجزیه و تحلیل فوریه

۵- تجزیه و تحلیل نوین

تجزیه و تحلیل مدار dc

این تجزیه و تحلیل دارای چند نوع می‌باشد که در همه انواع تمام خازنها مدار باز و تمام سلفها اتصال کوتاه در نظر گرفته می‌شوند. انواع این تجزیه و تحلیل عبارتند از :

الف - .op. این دستور (تجزیه و تحلیل) نقاط کار مدار را بدست می‌آورد. نقاط کار شامل ولتاژ گرهها، جریان شاخه‌ها و توان مصرفی المانها می‌باشد. علاوه بر اینها پارامترهای مدل سیگنال کوچک عناصر فعال نیز ارائه می‌گردد. این نتایج در `filename.Lis` می‌باشند. شکل کلی این دستور بصورت‌های زیر می‌باشد.

`.op`

`.op [format] [time] [format] [time]`

اگر در جلو این دستور چیزی نوشته نشود نقاط کار (ولتاژ، جریان، توان المانها و پارامترهای سیگنال کوچک عناصر فعال) را می‌دهد ولی با اضافه کردن کلماتی مثل دستور زیر می‌توان یکی از موارد نقاط کار را در زمان معین بدست آورد.

ب - .SENS. این دستور (تجزیه و تحلیل) حساسیت ولتاژها و یا جریان‌های مشخصی از مدار را نسبت به تغییرات پارامترهای مدار ارائه می‌کند. تغییر پارامترها با دستور `Param`. انجام می‌گیرد که بعد توضیح داده می‌شود. شکل کلی این دستور بصورت زیر می‌باشد .

`.SENS < یک یا چند ولتاژ یا جریان >`

مثال :

`.SENS V(9) V(4,3) V(17) I(VCC)`

ج - .TF. این دستور (تجزیه و تحلیل) مقدار تابع تبدیل سیگنال کوچک، مقاومت ورودی و مقاومت خروجی مدار را بدست می‌آورد. برای مثال اگر $V(in)$ ولتاژ ورودی و $V(out)$ ولتاژ خروجی باشد بهره ولتاژ A_v دارای تعریف زیر خواهد بود. این نتایج در `filename.Lis` می‌باشند.

$$A_u = \frac{\Delta V_o}{\Delta V_i} = \frac{V(out)}{V(in)}$$

برای بیان بهره ولتاژ فوق از دستور زیر استفاده می‌کنیم :

`.TF Vout Vin`

شکل کلی این دستور بصورت زیر می‌باشد.

`.TF Y X`

که X و Y ولتاژ یا جریان نقاط مختلف می‌باشند.

د - DC. با این دستور (تجزیه و تحلیل)، مشخصه انتقالی مدار بدست می‌آید. به این معنی که ورودی بطور dc بین دو حد مشخص تغییر می‌کند و خروجی اندازه‌گیری شده و سپس منحنی خروجی برحسب تغییرات ورودی ارائه می‌گردد. شکل ساده دستور DC، بصورت زیر است .

.DC Input Start Value Stop Value Increment Value
Start Value مقدار شروع، Stop Value مقدار نهائی و Increment Value مقدار گام افزایش ورودی می‌باشند .

مثال :

.DC Vin -5V 10V 0.25V

می‌توان دستور DC را به فرم زیر نیز نوشت .

مثال :

.DC < Oct or Dec> NP Start Value Stop Value

Oct or Dec نشان دهنده طرز تغییرات ورودی و NP معرف تعداد گامها در هر Octave یا Decade

می‌باشند. اگر دو ورودی در مدار وجود داشته باشند که هر دو بصورت DC در حوزه مشخصی تغییر کنند می‌توانیم برای هر دو آنها با یک دستور DC این تجزیه و تحلیل را انجام دهیم .

.DC In1 Start Value1 Stop Value1 Inc. Value1 In2 Start Value2 Stop Value2
+ Inc. Value2

.DC In1 <Oct or Dec> NP1 Start Val.1 Stop Val.1 In2 <Oct or Dec> NP2
+Start Val.2 Stop Val.2

مثال :

.Dc Iref dec 5 10nA 10uA Vtest dec 2 0.5 5

تجزیه و تحلیل ac (AC).

این تجزیه و تحلیل پاسخ فرکانسی را در یک پهنای باند مشخص از فرکانس بدست می‌آورد و برحسب درجه بندی شدن محور فرکانس به یکی از سه صورت کلی زیر نوشته می‌شود. برای این نوع تجزیه و تحلیل باید یک ورودی AC یا یک ورودی Mixed تعریف شود.

.AC LIN NP FStart FStop

.AC OCT NP FStart FStop

.AC DEC NP FStart FStop

LIN به معنی درجه بندی شدن محور فرکانس از FStart تا FStop بصورت خطی، OCT بصورت

لگاریتمی در مبنای هشت و DEC بصورت لگاریتمی در مبنای دهدهی می‌باشند.

NP (Number of points) : اگر تجزیه و تحلیل با LIN انجام می‌گیرد تعداد فرکانس‌هایی را نشان می‌دهد که تجزیه و تحلیل در آن نقاط انجام می‌گیرد. اگر از OCT یا DEC استفاده شود NP معرف تعداد فرکانس‌هایی است که در هر Octave و یا در هر Decade مورد بررسی قرار می‌گیرد. از تجزیه و تحلیل AC LIN برای پهنای باندهای باریک و از AC OCT برای پهنای باندهای متوسط و از AC DEC برای پهنای باند بزرگ استفاده می‌شود.

مثال :

.AC LIN 150 100Hz 300Hz	باند باریک
.AC OCT 10 100Hz 10kHz	باند متوسط
.AC DEC 10 100Hz 10MHz	باند عریض

تبصره : می‌توان با دستورهای زیر تجزیه و تحلیل AC را به ازای مقادیر مختلف یک ورودی یا به ازای دماهای مختلف انجام داد.

```
.AC LIN NP FStart Fstop Sweep Input <Oct or Dec> NP1 Start Value Stop Value
.AC OCT NP FStart FStop Sweep Input <Oct or Dec> NP1 Start Value Stop Value
.AC DEC NP FStart Fstop Sweep Input <Oct or Dec> NP1 Start Value Stop Value
```

مثال :

```
.AC dec 10 100 100k sweep Iref dec 2 10n 10u
.AC dec 10 100 100k sweep temp 25 50 75 100
```

تجزیه و تحلیل گذرا (TRAN).

در این تجزیه و تحلیل پاسخ گذرای یک مدار در حوزه زمان برای ورودی معینی در حوزه زمان مشخص می‌گردد. دستور TRAN بصورت دو شکل زیر می‌باشد. برای این نوع تجزیه و تحلیل باید یک ورودی Sin، Pulse، نمائی، پاره خطی یا Mixed تعریف شود.

```
.TRAN TSTEP TSTOP <UIC>
.TRAN TSTART TSTOP TSTEP <UIC>
```

TSTEP زمانی است که هر بار به زمان قبلی اضافه شده و در آن لحظه مدار تجزیه و تحلیل می‌گردد.

TSTOP زمان انتهای تجزیه و تحلیل است.

TSTART زمان شروع تجزیه و تحلیل است و اگر نوشته نشود برنامه به طور اتوماتیک آن را صفر فرض

می‌کند.

UIC از حروف اول کلمات Use Initial Conditions اخذ شده است. اگر خازنها و سلفها دارای شرایط اولیه باشند و بخواهیم در تجزیه و تحلیل گذرا این شرایط اولیه در نظر گرفته شوند باید از کلمه UIC در داخل دستور TRAN استفاده شود.

مثال :

.TRAN 5uS 1mS

یعنی مدار را از صفر تا ۱ msec و در فواصل ۵ usec ای تجزیه و تحلیل گذرا کن.

.TRAN 5uS 1mS 200uS

یعنی مدار را بین زمان ۵ uSec تا ۱ mSec و به فواصل ۲۰۰ uSec تجزیه و تحلیل گذرا کن.

مثال : فرض کنید یک خازن و یک سلف با شرایط اولیه در مدار وجود دارد در این صورت دستور

TRAN به شکل زیر نوشته می شود.

CX 3 4 10uF IC=-2.5V
LY 10 0 0.5uH IC= 24.3mA
.TRAN 5uS 1mS UIC

تبصره : برای سلفها و خازنها فقط در رابطه با تجزیه و تحلیل گذرا باید IC=X (شرایط اولیه) را مشخص نمود و در رابطه با بقیه تجزیه و تحلیلها احتیاجی به این کار نیست .

تبصره : لازم نیست شرایط اولیه مدار از طریق تعریف در سلف و خازن نوشته شود. می توان با دستور IC به هر کدام از گره های مدار شرایط اولیه دلخواهی را نسبت داد و از برنامه خواست با توجه به آن شرایط اولیه، مدار را تجزیه و تحلیل کند. شکل کلی دستور IC به صورت زیر است .

.IC V(1)=V₁ V(2)=V₂ V(3,4)=V₃ ...

تبصره : اگر از دستور IC استفاده شود دیگر نباید در دستور TRAN بخش UIC نوشته شود.

تبصره : می توان با دستورهای زیر تجزیه و تحلیل Transient را به ازای مقادیر مختلف یک ورودی یا به ازای دماهای مختلف انجام داد.

.Tran TStep TStop <UIC> Sweep Input <Oct or Dec> NP Start Value Stop Value

.Tran TStart TStop TStep <UIC> Sweep Input <Oct or Dec> NP Start Val. Stop Val.

مثال :

.Tran 5uS 1mS UIC sweep Iref dec 2 10n 10u

.Tran 5uS 1mS UIC sweep temp 25 50 75 100

تجزیه و تحلیل فوریه (FOUR).

خروجی های بدست آمده در تجزیه و تحلیل گذرا جداگانه به صورت جداولی موجودند. این نمونه های اطلاعات را می توان برای بدست آوردن یک سری فوریه از سیگنال خروجی مورد استفاده قرار داد. یک سیگنال پریودیک را می توان توسط رابطه سری فوریه به صورت زیر نشان داد.

$$V(w) = C_0 + \sum_{n=1}^{\infty} C_n \sin(nw + f_n)$$

که در آن $w = 2\pi ft$ (f فرکانس هارمونی اصلی) ، C_0 مقدار متوسط موج و C_n دامنه هارمونی n ام و ϕ_n فاز هارمونی n ام می باشد. HSpice با استفاده از نتایج تحلیل گذرا ضرایب ۹ هارمونیک اول از سری فوریه را بدست می آورد. شکل کلی دستور Four. به صورت زیر است :

.Four FREQ X₁ X₂ ... X_n

X_i ها ولتاژها و جریانهای خروجی هستند که مایلیم سری فوریه آنها را بدانیم و FREQ فرکانس هارمونیک اول می باشد.

مثال :

.Four 100kHz V(2,3) V(3) I(R1) I(VIN)

تجزیه و تحلیل نویز (NOISE).

همانطوریکه می دانیم مقاومتها و نیمه هادی ها در مدار Noise تولید می کنند و توان این نویز تابعی از پهنای باند مدار است. در HSpice ، تجزیه و تحلیل Noise همراه با تجزیه و تحلیل AC میسر است. مقادیر نویز در خروجی تحلیل AC می باشد. برای هر فرکانس تمام منابع نویز مدار محاسبه شده و اثر مجموعه آنها در خروجی به صورت کل نویز خروجی وارد می گردد. عبارت کلی برای تجزیه و تحلیل نویز به شکل زیر است.

.NOISE V(N+,N-) Source [M]

که در این رابطه V(N+,N-) ولتاژ خروجی بین دو گره N+ و N- است. البته خروجی می تواند بین یک گره و زمین باشد که در این صورت V(N) را خواهیم داشت. Source نام یک منبع ولتاژ و یا منبع جریان مستقل در ورودی مدار است که نویز به صورت ولتاژ V_n^2 و جریان I_n^2 نسبت به آن منبع محاسبه می گردد.

مثال :

.Noise V(4,5) Vin

.Noise V(6) Iin

چاپ و رسم خروجی‌های مورد نظر در تجزیه و تحلیل‌های مختلف

دستوراتی که به کمک آنها می‌توان نتایج تجزیه و تحلیل‌های پنجگانه را مشاهده نمود و خروجی‌های مورد نظر را بدست آورد یا منحنی‌های مورد نظر را رسم نمود شامل موارد زیر می‌باشند.

.PRINT ، .PROBE ، .PLOT

.PRINT برای چاپ نتایج بصورت گرافیکی و عددی بکار می‌رود این اعداد با توجه به نوع تجزیه و تحلیل بصورت زوج مرتب (مثلا در تجزیه و تحلیل گذرا ولتاژ یا جریان خواسته شده در دستور **.PRINT** به همراه زمان) ارائه می‌گردند. از این اعداد می‌توان بعنوان داده (مثلا در Matlab) استفاده کرد و پردازشهای مورد نظر را انجام داد.

.PROBE برای رسم منحنی تغییرات پارامترهای موردنظر بکار می‌رود.

.PLOT همان نقش **.PROBE** را دارد منتهی با کیفیت و امکانات بسیار پائین و به همین دلیل در مورد آن بحثی نخواهیم داشت.

دستور **.PRINT**

نتایج تجزیه و تحلیل‌های **DC**، **AC**، گذرا و **NOISE** را می‌توان با کمک دستور **.PRINT** بصورت گرافیکی و عددی بدست آورد. شکل کلی نوشتن این دستورات بصورت‌های زیر است.

.PRINT DC [متغیرهای خروجی مورد نظر]

.PRINT AC [متغیرهای خروجی مورد نظر]

.PRINT TRAN [متغیرهای خروجی مورد نظر]

.PRINT NOISE [متغیرهای خروجی مورد نظر]

مثال :

.PRINT DC V(2) V(3,5) V(R1) VCE(Q3)

.PRINT AC V(1) V(2) I(R2) I(C1)

.PRINT TRAN V(in) V(out)

تبصره ۱ : با یک **.PRINT** فقط اطلاعات مربوط به ۳۲ متغیر را می‌توان محاسبه و ضبط نمود و برای

محاسبه و ضبط متغیرهای بیشتر باید به تعداد دستوره‌های **.PRINT** اضافه نمود.

تبصره ۲: می‌توان در دستور PRINT بجای "متغیرهای خروجی مورد نظر"، یک یا چند پارامتر دیگر را بصورت تابعی از ولتاژ یا جریانهای نقاط دیگر تعریف کرد. این پارامتر بعد از اجرا، در داخل خروجی تحلیل مربوطه به عنوان یک متغیر با نوع Params ظاهر می‌شوند که با کلیک کردن روی آن می‌توان شکل موج توابع تعریف شده را دید. البته تعریف کردن توابع (یا رسم تابعی از چند متغیر) بعد از اجرا کردن، در خود محیط Avan waves قابل اجراست و نیاز به تعریف کردن در این قسمت نیست. می‌توان برای توابع، اسم نیز در نظر گرفت مثل gain در مثال زیر.

```
.PRINT TRAN par('(20*log10(v(out)/v(in2))))'
```

```
.PRINT AC gain = PAR('v(3)/v(2)') PAR('v(4)/v(2)')
```

دستور PROBE

برای دیدن شکل موجهای خواسته شده در تجزیه و تحلیل‌های DC ، AC و TRAN از دستور PROBE استفاده می‌شود. اگر بخواهیم تمام متغیرها را در تمام تجزیه و تحلیل‌ها مشخص شده ببینیم این دستور دارای شکل PROBE خواهد بود. ولی اگر بخواهیم بعضی از متغیرها را در تجزیه و تحلیل خاصی مشاهده کنیم شکل کلی این دستور بصورت‌های زیر نوشته می‌شود. در اینجا نیز می‌توان توابعی را بصورت پارامتر تعریف کرد.

```
.Probe DC [ متغیرهای خروجی مورد نظر ]
```

```
.Probe AC [ متغیرهای خروجی مورد نظر ]
```

```
.Probe TRAN [ متغیرهای خروجی مورد نظر ]
```

```
.Probe NOISE [ متغیرهای خروجی مورد نظر ]
```

مثال :

```
.Probe DC V(2) V(3,5) V(R1) VCE(Q3)
```

```
.Probe AC V(1) V(2) I(R2) I(C1)
```

```
.Probe TRAN V(in) V(out)
```

```
.Probe TRAN par('(20*log10(v(out)/v(in2))))'
```

```
.Probe AC gain = PAR('v(3)/v(2)') PAR('v(4)/v(2)')
```

آشنایی با چند دستور مفید دیگر

SUBCKT: اگر یک مدار مشخص بعنوان یک بخش کوچک از یک مدار بزرگ چندین بار در جاهای مختلف یک مدار بزرگ تکرار شود می‌توان با دستور **SUBCKT** یک زیر برنامه برای آن بخش تکرار شونده نوشت و سپس از آن زیر برنامه در نوشتن کل برنامه برای مدار استفاده کرد (مانند Subroutine در برنامه). زیر برنامه **SUBCKT** دارای فرمتی به صورت زیر است.

نوشتن زیر برنامه

شماره‌های چند گره < یک نام دلخواه برای زیر برنامه > **SUBCKT**

توصیف زیر برنامه: برنامه مربوط به بیان المان‌های زیر مدار

ENDS < همان نام دلخواه بالایی >

نکته ۱: شماره گره‌هایی که در جلو دستور **SUBCKT** و یا در داخل زیر برنامه نوشته می‌شوند گره‌های محلی می‌باشند یعنی می‌توان همان شماره‌ها را در داخل برنامه اصلی یا در زیر برنامه دیگری استفاده کرد.

نکته ۲: زیر برنامه فقط شامل عناصر مدار می‌باشد و به جز دستور **MODEL** از هیچ دستور نقطه‌دار نباید در زیر برنامه استفاده شود.

نکته ۳: زیر برنامه را می‌توان در هر کجای برنامه اصلی قرار داد یا می‌توان در فایل دیگر نوشت و در داخل برنامه اصلی آن را صدا کرد. اگر زیر برنامه در فایل دیگری نوشته شود باید در اول برنامه اصلی با استفاده از دستور **include** آن را فراخوانی کرد. برای مثال اگر زیر برنامه‌ای به اسم **Amplifier.sp** در شاخه **E:\Project** نوشته شده، باید آنرا بصورت زیر در برنامه اصلی فراخوانی کرد.

include "E:\Project\Amplifier.sp"

مثال:

```
.SUBCKT STMA 5 6 0
R1 1 2 1k
R2 2 0 1k
RD 1 3 10k
C1 5 2 10u
C2 3 6 1u
MQ1 3 2 0 0 MOS1 W=1u L= 1u
VCC 1 0 5dc
.MODEL MOS1 NMOS(VT0= 1 BETA= 5E- 4 RD= 4)
.ENDS STMA
```

طرز استفاده از زیر برنامه

پس از نوشتن زیر برنامه (و فراخوانی آن در صورت نوشته شدن در فایل دیگر) می توانیم با دستوری به شکل زیر از آن استفاده کنیم.

نام انتخاب شده برای زیر برنامه < شماره چند گره متناظر با گره های زیر برنامه > X(name)

مثال :

```
X1      6      7      0      STMA
X2      4      5      0      STMA
```

دستور Data.

با این دستور می توان برای یک پارامتر چند تا داده تخصیص داد و بعد به ازای این داده ها مدار را تجزیه و تحلیل کرد. یا به عبارتی این دستور برای ساختن Table بکار می رود. فرم این دستور بصورت زیر است.

```
.DATA dataname pnam1 pnam2 pnam3 ...
```

مقادیر برای pnam3 و pnam2 و pnam1 (با ترتیب)

```
.ENDDATA
```

طرز استفاده :

مثال :

```
.AC dec 10 100 10meg SWEEP DATA = dataname
```

```
.TRAN 1n 10n SWEEP DATA = dataname
```

مثال کلی:

```
.TRAN 1n 100n SWEEP DATA = devinf
```

```
.AC DEC 10 1hz 10khz SWEEP DATA = devinf
```

```
.DC TEMP -55 125 10 SWEEP DATA = devinf
```

```
.DATA devinf width length thresh cap
      + 50u 30u 1.2v 1.2pf
      + 25u 15u 1.0v 0.8pf
      + 5u 2u 0.7v 0.6pf
```

```
.ENDDATA
```

در مثال فوق تحلیل Tran، AC و DC (به ازای هر دما) سه بار (تعداد داده ها) انجام می گیرند.

برخی دستورات اضافی:

One-Port Network
.NET input

Two-Port Network
.NET output input

این دستور برای انجام تحلیل شبکه و یافتن پارامترهای تک-قطبی و دو-قطبی شبکه می باشد.
از جمله امپدانس یا ادمیتانس ورودی یا پارامترهای Z, Y, S, H شبکه دو-قطبی است.

Input: هر منبع ولتاژ یا جریان ac

Output: ولتاژ یا جریان هر گره

نکته: این تحلیل باید به همراه تحلیل ac صورت گیرد.

و برای دیدن پارامترها پس از تحلیل از print استفاده می کنیم.

```
.PRINT AC Z11(R) Z12(R) Y21(I) Y22 S11  
+S11(DB) Z11(T)
```

```
.param par1=val1 par2=val2 ...
```

این دستور برای تعریف پارامترهای به کار رفته در مدار اصلی به کار می روند.

```
.op
```

این دستور به محاسبه نقطه کار dc مدار می پردازد.

```
.pz output input
```

برای انجام تحلیل صفر و قطب که بیان کننده تابع تبدیل یک شبکه مستقل از زمان است.

Output: هر متغیر خروجی مثل ولتاژ گره یا جریان شاخه

Input: منبع ولتاژ یا جریان مستقل

$$H(s) = \frac{a_0 \cdot (s+z_1)(s+z_2)\dots(s+z_l)\dots(s+z_m)}{b_0 \cdot (s+p_1)(s+p_2)\dots(s+p_j)\dots(s+p_m)}$$

پس از انجام این دستور صفرها و قطبها بر حسب (rad/sec) و (hertz) لیست می شود.

```
.FFT <output var> <np=var> *****
```

برای انجام تحلیل فوریه گسسته در زمان صورت می گیرد.

Output var: هر متغیر خروجی مثل ولتاژ گره یا جریان شاخه

Np: تعداد نقاط کل تحلیل که باید توانی از ۲ باشد.

- .option post ***** برای اینکه نتایج شبیه سازی به صورت منحنی دیده شود
- .option accurate ***** برای مدارات با رنج دینامیکی وسیع در تحلیل گذرا
- .option fast ***** افزایش سرعت تحلیل در مدارات دیودی و ترانزیستوری پیچیده
- .option list ***** چاپ لیستی از قطعات، منابع، مقادیر آنها و...
- .option method=gear/trap روش ریاضی همگرایی اعداد در حوزه زمان را بیان میکند.
Trap: معمولا حجم محاسبات پایین و همگرایی خوبی دارد.
Gear: برای مدارات پیچیده غیر خطی
- .option nomod ***** برای اینکه پارامترهای مدل را به خروجی نبرد

دستورات اندازه گیری (Measurement):

این دستورات برای اندازه گیری از روی شبیه سازی هایی که نرم افزار با موفقیت انجام داده است. نکته: در این دستورات به جای `measure` می توان از `meas` استفاده کرد.

*MEASURE (Rise, Fall, and Delay Measurements)

`.MEASURE <DC | AC | TRAN> result TRIG ... TARG ...`
پس از `meas` باید نوع تحلیل را مشخص کرد و بعد آن یک نام دلخواه برای متغیری که می خواهیم اندازه گیری شود.

```
TRIG var VAL=val < TD=td > < CROSS=c | LAST >
+ < RISE=r | LAST > < FALL=f | LAST >
+ <TRIG AT=time>
TARG var VAL=val < TD=td > < CROSS=c | LAST >
+ < RISE= r | LAST > < FALL=f | LAST> <REVERSE>
+ <TRIG AT=time>
```

`Trig` یعنی از کجای یک نمودار اندازه گیری را شروع کن. `var` نام متغیر مثل ولتاژ گره یا جریان شاخه ی است و `val` مقدار آن متغیر می باشد.

`Targ` هم میگوید که تا کجا به اندازه گیری ادامه دهد. (توضیح بقیه گزینه ها در مثالها بیان می شود)

*MEASURE (FIND and WHEN)

```
.MEASURE <DC | AC | TRAN> result
+ WHEN out_var=val <TD=val> <FROM=val> <TO=val>
+ < RISE=r | LAST > < FALL=f | LAST >
+ < CROSS=c | LAST >
.MEASURE <DC | AC | TRAN> result FIND out_var1
+ WHEN out_var2=val < TD=val > <FROM=val> <TO=val>
+ < RISE=r | LAST > < FALL=f | LAST > < CROSS=c | LAST >
(refer for more syntax to hspice_cmdref page 115)
```

*MEASURE (parametric)

```
.MEASURE <DC | TRAN | AC> result PARAM='equation'
```

برای محاسبه یک پارامتر خاص بر اساس نتایج اندازه گیری های قبلی
نکته: در داخل عبارت equation نمی توان جریان شاخه یا ولتاژ گره ی را قرار داد.

***MEASURE (Average, RMS, MIN, MAX, INTEG, and PP)**

.MEASURE <DC|AC|TRAN> result func out_var


Func باید avg, rms, min, max, integ, pp باشد تا میانگین، rms، کمینه،

بیشینه، انتگرال و مقدار پیک تا پیک متغیر خروجی را اندازه گیری کند.

مثالها:

Band Pass Filter: در این مثال به توصیف مدار یک فیلتر میان گذر می پردازیم ، سپس با انجام تحلیل ac پاسخ آن را در باند فرکانسی مشاهده میکنیم و در آخر مقادیر f_0, f_1, f_2, BW, Q را برای این فیلتر میان گذر اندازه گیری می کنیم.

Schematic View:



فایل netlist ورودی:

```

#BPF#
*****ANALYSIS
.OPTIONS POST
.ac dec 20 1 100x
.pz v(out) vin
*****circuit
Vin in 0 ac 1
R1 in 1 R
R2 2 0 R
R3 out 1 R
R4 out 3 RB
R5 3 0 RA
C1 1 2 C
C2 1 0 C
E_opamp out 0 Vol='1e5*V(2,3)'
*****
.param R=4.5k C=500p k=3.8 RA=1k RB='RA*(k-1)' $line16
+m0=1.4125375 $line17
*****meas $line18
.meas ac Vpo max VM(out) $line19
.meas ac F0 when VM(out)='Vpo' $line20
.meas ac F1 trig at=1 targ VM(out) val='(Vpo)/m0' cross=1 $line21
.meas ac F2 trig at=1 targ VM(out) val='(Vpo)/m0' cross=2 $line22
.meas ac BW param='F2-F1' $line23

```

.meas ac QF param='F0/BW'

\$line24

\$line25

.END

\$line26

Line1: این خط را نرم افزار به عنوان تیتر می شناسد وکل آن را به کپی میکند.
خطوط ۲ و ۶ و ۸ و ۲۵ چون با علامت * شروع شده اند نرم افزار آنها را ترجمه نمی کندو در netlist ها برای تفکیک بخشهای مختلف به کار می روند.
Line3: این دستور در اینجا از ۱ گزینه تشکیل شده که: post برای ارسال نتایج شبیه سازی به خروجی گرافیکی می باشد.

Line4: دستور تحلیل ac از فرکانس 1hz تا 100megahz با دقت هر دهه ۲۰ نقطه می باشد.
Line5: جهت انجام تحلیل صفر و قطب و ارائه تابع تبدیل از ولتاژ گره out به منبع vin می باشد.


Line7: منبع ولتاژ به نام vin با مقدار دامنه ac ۱ ولت
خطوط ۸ تا ۱۴ به توصیف قطعات مختلف مدار می پردازد که مقادیر آنها به صورت پارامتری بیان شده است.

Line15: منبع ولتاژ وابسته به نام E_opamp بین گره های out و زمین که به صورت عبارت داخل " با گره های ۲ و ۳ رابطه دارد.
Line16: این خط به تعریف پارامترهایی که در کل فایل به آن ارجاع داده شده می پردازد و مقادیر عددی را به آنها نسبت می دهد. خط ۱۷ هم ادامه خط ۱۶ می باشد که با علامت + شروع شده.

خطوط ۱۹ تا ۲۴ برای اندازه گیری مقادیر شبیه سازی است :
Line19: یعنی بیشترین مقدار دامنه ولتاژ گره out در طول تحلیل ac راباید و نام آن را Vpو بگذارد.


Line20: یعنی فرکانسی که در آن دامنه ولتاژ خروجی Vpو (بیشینه) شود را F0 بنامد.
Line21,22: یعنی جایی که دامنه ولتاژ خروجی 3db- شود برای اولین بار فرکانس f1 و برای دومین بار f2 بنامد.

Line23,24: برای محاسبه پهنای باند فیلتر و ضریب کیفیت فیلتر بر اساس اندازه گیری های بالا از شکل پارامتری دستور .meas استفاده شده.
Line26: با نوشتن دستور .End می گویم که فایل مدار پایان یافته است.



Current Source: در این مثال میخواهیم ضمن شبیه سازی یک منبع جریان تغییرات جریان خروجی را با تغییر پارامتری از مدار بررسی کنیم ، همچنین به مقاومت و خازن خروجی منبع جریان را در فرکانس های مختلف مشاهده کنیم.

Schematic View



فایل netlist ورودی:

```
#CMOS_C.S#
*****ANALYSIS
.OPTIONS POST nomod method=GEAR          $1
.op                                         $2
.dc Vt 0 10 0.1                           $3
.probe dc I1(m2)                           $4
.ac dec 15 1k 1g
.net Vt                                     $5
.probe ac yin(r) yin(i)                   $6
.probe ac Cout=par('yin(i)/(2*pi*hertz)') $7
+Rout=par('1/yin(r)')                     $8
*****circuit
M1  1 1 0 0  cmosn L=0.5U W=5U           $9
M2  2 1 3 3  cmosn L=0.5U W=5U
IREF vdd 1 iref
VDD vdd 0vdd
Vt  2 0 Vt ac 1                            $10
Rs  3 0 Rs
*****param
.param      iref=400u vdd=5 Rs=1k Vt=5
+pi=3.14159265
*****
.inc  'cmos.txt'                            $11
.END
```

Line1: nomod در دستور option. برای این است پارامترهای مدل ماسفت را در خروجی نشان ندهد.

Line2: ارائه نقطه کار ترانزیستورها و ولتاژ گره ها در خروجی

Line3: جاروب کردن پارامتر V_t از ۰ ولت تا ۱۰ ولت با فاصله ۰.۱.

Line4: یعنی جریان درین ماسفت M2 را در خروجی نشان دهد.

Line5: این دستور بیان می کند که تحلیل شبکه را با پورت ورودی V_t انجام دهد و امپدانس ورودی یا ادمیتانس ورودی را محاسبه کند.

Line6: یعنی بخش حقیقی و موهومی ادمیتانس ورودی را در خروجی نمایش بده.



Line7,8: یعنی پارامترهای C_{out} , R_{out} را محاسبه و در خروجی نمایش بده.

Line9: ماسفت M1 که درین، گیت، سورس و بدنه آن به ترتیب بین گره های ۱، ۱، ۰، ۰، می باشد. نام مدل آن cmosn است که در فایل cmos.txt به netlist الحاق شده است.


L و W هم طول و عرض کانال آن میباشد.

Line10: منبع ولتاژ V_t که مقدار dc آن به صورت پارامتری تعریف شده و مقدار ac آن هم ۱ قرار داده شده.

Line11: این دستور فایل cmos.txt را که در پوشهء فایل sp. قرار دارد را به ادامه netlist پیوست می کند.




:BJT Amplifier



فایل netlist ورودی:

```
#Bjt_exp#
.option post nomod accurate          $1
.ac dec 20 1 1g
.op
.PZ  V(out) Vs
.tran 100n 100u                      $2
*****Circuit
Q1  c b e      my_model              $3
Rc  c vcc     2.2k
R1  vcc b     33k
R2  b 0       10k
Re  e 0       330
Rs  s in      600
Cb1 in b      cby_pass
Cb2 c out     cby_pass
RL  out 0     40k
CL  out 0     CL
Vcc vcc 0     vcc
Vs  s 0      sin(0,1m,50k) ac(1m,0)  $4
*****Param
.param cby_pass=100n CL=10p Vcc=5 m0=1.4125375
.model my_model npn                  $5
+ vje=0.65 vjc=0.75 rb=100 VAF=75
+ mjc=0.3 Bf=100 cjc=8pF CJE=20pF
*****Meas
.meas ac Amid max VM(out)
.meas ac Fhi when VM(out)='Amid/m0' cross=2
.meas ac Flo when VM(out)='Amid/m0' cross=1
*****
.END
```


- Line1: گزینه accurate در دستور option. برای افزایش دقت در تحلیل گزرا می باشد.
- Line2: دستور تحلیل حوزه زمان با گام 100ns تا زمان 100us
- Line3: ترانزیستور BJT با نام Q1 که کلکتور و بیس و امیتر آن بین گره های c,b,e قرار گرفته اند و نام مدل آن my_model می باشد.
- Line4: منبع ولتاژ Vs با مقدار ac ، دامنه 1mv و با فاز 0 درجه و مقدار حوزه زمان سینوسی با آفست 0 ولت و دامنه 1mv و فرکانس 50khz
- Line5: دستور model. برای معرفی مدل ترانزیستور Q1 به نام my_model است.



نکته: نتایج دستورات اندازه گیری در فایل notepad که پس از شبیه سازی توسط نرم افزار تولید می شود قابل مشاهده است. (گزینه Edit LL)

نکته: اگر به علت وجود خطایی در فایل .sp تولید شده توسط کاربر نرم افزار شبیه سازی نکرد ، علت خطا را می توان با مشاهده فایل notepad بالا شناسایی کرد.

:CMOS Amplifier



فایل netlist ورودی:

```
#CMOS_Amp#
*****ANALYSIS
.OPTION POST nomod method=GEAR      accurate dcon= 1      $1
.op
.ac dec 20 1 100g sweep phi poi 1 0      $2
.ac dec 20 1 100g sweep phi poi 1 180    $3
.probe ac gain=par('db(vm(out))')
.tran 100n 100u
.four 100k V(out)                        $4
.pz V(out) vinp
*****circuit
M1 1 inp 2 2 cmosn L=0.5U W=5U
M2 3 inn 2 2 cmosn L=0.5U W=5U
M3 1 1 vdd vdd cmosp L=0.5u W=1u
M4 3 1 vdd vdd cmosp L=0.5u W=5u
M5 vdd 3 out out cmosn L=0.5u W=2u
M6 4 4 vss vss cmosn L=0.5u W=1u
M7 2 4 vss vss cmosn L=0.5u W=1u
M8 out 4 vss vss cmosn L=0.5u W=2u
Iref vdd 4 iref
vdd vdd 0 5
vss vss 0 -5
vinp inp 0 ac(0.5,0) sin(0,1m,100k)
vinn inn 0 ac(0.5,phi)
*****param
.param iref=10u pi=3.14159265 phi=0
*****
.inc 'cmos.txt'
.end
```

Line1: برای همگرایی بهتر به جواب $dcon=1$ قرار می دهند. البته خود HSPICE در صورت عدم همگرایی آن را برابر ۱ قرار می دهد.

Line2,3: دستور تحلیل ac از فرکانس 1hz تا 100Ghz با دقت هر دهه ۲۰ نقطه. همچنین ادامهء دستور بیانگر تحلیل ثانویه با تغییر پارامتر phi می باشد که یک بار برابر 0 درجه و یک بار 180 درجه قرار گرفته است.

Line4: دستور تحلیل فوریه روی ولتاژ گره out در حوزه زمان با فرکانس مرکزی 100khz است. این دستور هارمونیک های بالاتر ورودی را هم در خروجی محاسبه می کند تا مقدار کل اعوجاج هارمونیک سیستم مشخص شود.

نتیجهء تحلیل فوریه را در فایل notepad خروجی مشاهده می کنیم:

1***** HSPICE -- A-2008.03 32-BIT (Feb 26 2008) winnt *****

#cmos_amp#

tnom= 25.000 temp= 25.000

fourier components of transient response v(out)

dc component = 2.655E+00


harmonic frequency fourier normalized phase normalized

no (hz) component component (deg) phase (deg)

1	100.0000k	138.6932m	1.0000	-151.1564m	0.
2	200.0000k	196.7727u	1.4188m	-93.1006	-92.9494
3	300.0000k	5.3292u	38.4241u	-48.8121	-48.6609
4	400.0000k	11.7162u	84.4754u	-124.6280	-124.4769
5	500.0000k	24.2647u	174.9525u	47.8343	47.9855
6	600.0000k	10.0488u	72.4538u	-88.0363	-87.8851
7	700.0000k	47.2370u	340.5865u	51.1175	51.2687
8	800.0000k	4.1643u	30.0253u	-101.2896	-101.1384
9	900.0000k	10.3221u	74.4242u	-13.3210	-13.1699

total harmonic distortion = 147.6414m percent

نتیجه تحلیل ac و دامنه و فاز سیگنال وجه تفاضلی و مشترک را بر حسب db ملاحظه می کنیم.



نتیجه تحلیل حوزه زمان و سیگنال های ورودی و خروجی:

